

Práctica 6: Funciones de varias variables

Ejemplo 1. Límites de funciones

Para hallar el límite de la función $f(x,y) = \frac{x^2 \cdot y}{x^2 + y^2}$ en un punto se ejecuta el siguiente comando:

```
> f:=(x,y)-> x^2*y/(x^2+y^2);  
> Limit(f(x,y),{x=2,y=-1})=limit(f(x,y),{x=2,y=-1});
```

En ocasiones, por ejemplo en el punto (0, 0), maple no sabe dar un valor para el límite:

```
> Limit(f(x,y),{x=0,y=0})=limit(f(x,y),{x=0,y=0});
```

En estos casos, se puede recurrir a la representación gráfica para estudiar lo que ocurre:

```
> plot3d(f(x,y), x=-1..1, y=-1..1,numpoints=1000, style=patch,  
axes=framed);
```

La gráfica indica que, a pesar de presentar un agujero en el origen, el límite debería ser 0, ya que la función está próxima a 0 cuando x e y están próximos a cero. Los límites iterados y direccionales confirman nuestra hipótesis:

```
> Limit(Limit(f(x,y),y=0),x=0)=limit(limit(f(x,y),y=0),x=0);  
> Limit(Limit(f(x,y),x=0),y=0)=limit(limit(f(x,y),x=0),y=0);  
> Limit(f(x,k*x),x=0)=limit(f(x,k*x),x=0);
```

Nuestra hipótesis se confirma también al elegir otras direcciones:

```
> Limit(f(x,k*x^2),x=0)=limit(f(x,k*x^2),x=0);  
> Limit(f(x,k*sqrt(x)),x=0)=limit(f(x,k*sqrt(x)),x=0);
```

Para confirmarla, se recurre a coordenadas polares:

```
> f(rho*cos(theta),rho*sin(theta));  
> simplify(%);  
> maximize((cos(theta))^2*sin(theta),theta=0..2*Pi);  
> Limit(rho,rho=0)=limit(rho,rho=0);
```

de donde se concluye que el límite es cero.

Ejemplo 2. Funciones sin límite puntual

La función $f(x,y) = \frac{x \cdot y}{x^2 + y^2}$ no tiene límite en (0, 0).

```
> f:=(x,y)-> x*y/(x^2+y^2);
```

Maple no sabe hallarlo:

```
> Limit(f(x,y),{x=0,y=0})=limit(f(x,y),{x=0,y=0});
```

Los límites iterados coinciden, pero no así los direccionales:

```
> Limit(Limit(f(x,y),y=0),x=0)=limit(limit(f(x,y),y=0),x=0);  
> Limit(Limit(f(x,y),x=0),y=0)=limit(limit(f(x,y),x=0),y=0);  
> Limit(f(x,k*x),x=0)=limit(f(x,k*x),x=0);
```

de donde se deduce que el límite no existe, lo que se puede confirmar gráficamente:

```
> plot3d(f(x,y), x=-1..1, y=-1..1,numpoints=1000, style=patch,  
axes=framed);
```

La función $f(x,y) = \frac{x^2 \cdot y}{x^4 + y^2}$ no tiene límite en (0, 0).

```
> f:=(x,y)-> x^2*y/(x^4+y^2);
```

Maple no sabe hallarlo:

```
> Limit(f(x,y),{x=0,y=0})=limit(f(x,y),{x=0,y=0});
```

Los límites iterados y direccionales coinciden, pero no así los direccionales:

```
> Limit(Limit(f(x,y),y=0),x=0)=limit(limit(f(x,y),y=0),x=0);  
> Limit(Limit(f(x,y),x=0),y=0)=limit(limit(f(x,y),x=0),y=0);
```

```
> Limit(f(x,k*x),x=0)=limit(f(x,k*x),x=0);
```

pero hay direcciones en las que no coincide:

```
> Limit(f(x,x^2),x=0)=limit(f(x,x^2),x=0);
```

Por tanto, no existe el límite como se puede confirmar gráficamente:

```
> plot3d(f(x,y), x=-1..1, y=-1..1,numpoints=1000, style=patch, axes=framed);
```

Ejemplo 3. Derivadas parciales

Se considera la función $f(x,y) = x^2 \cdot \sin(x \cdot y)$. Para hallar sus derivadas parciales, se procede así:

```
> f:=(x,y)->x^2*sin(x*y);
> Diff(f(x,y),x)=diff(f(x,y),x);
> Diff(f(x,y),y)=diff(f(x,y),y);
> Diff(f(x,y),x,x)=diff(f(x,y),x,x);
> Diff(f(x,y),y,y)=diff(f(x,y),y,y);
> Diff(f(x,y),x,y)=diff(f(x,y),x,y);
> Diff(f(x,y),y,x)=diff(f(x,y),y,x);
> Diff(f(x,y),x,x,y)=diff(f(x,y),x,x,y);
```

Para hallar el gradiente:

```
> with(VectorCalculus):
> Gradient(f(x,y), [x,y]);
```

o también:

```
> Del(f(x,y), [x,y]);
```

La derivada direccional se calcula así (no es necesario normalizar el vector):

```
> DirectionalDiff(f(x,y), <1,1>, [x,y]);
> DirectionalDiff(f(x,y), <2,2>, [x,y]);
> simplify(%);
```

Para calcular las derivadas parciales, direccionales o gradiente en un punto, se procede así:

```
> eval(diff(f(x,y),x), [x=2,y=-1]);
> eval(diff(f(x,y),y,y), [x=Pi,y=1]);
> eval(diff(f(x,y),x,y), [x=1,y=Pi]);
> Gradient(f(x,y), [x,y]);
> eval(%,[x=1,y=Pi]);
> DirectionalDiff(f(x,y), <2,2>, [x,y]);
> eval(%,[x=1,y=Pi]);
```

O, mejor, se convierten primero en función y luego se evalúa en el punto deseado:

```
> DX:=unapply(diff(f(x,y),x), [x,y]);
> DX(2,-1);
> DXY:=unapply(diff(f(x,y),x,y), [x,y]);
> DXY(1,Pi);
> G:=unapply(Gradient(f(x,y), [x,y]), [x,y]);
> G(1,Pi);
```

Para hallar el hessiano de una función y evaluarlo en un punto, se hace así:

```
> Hessian(x^2+x*y^2, [x,y]);
> H:=unapply(%,[x,y]);
> H(-1,3);
> with(LinearAlgebra):
> Determinant(H(-1,3));
```

Ejemplo 4. Polinomios de Taylor

Para hallar el polinomios de Taylor se procede así:

```
> mtaylor(sin(x^2+y^2), [x=-1,y], 3);
```

```
> mtaylor(ln(x/y), [x=1,y=1], 3);
> mtaylor(x*cos(x+y), [x,y], 7);
```

Ejemplo 5. Extremos relativos

Para hallar los extremos relativos de la función $f(x,y) = -x^3 + 4xy - 2y^2 + 1$ se comienza hallando los puntos críticos:

```
> restart;
> f:=(x,y)->x^3+x^2*y+y^2+2*y;
> eq1:=diff(f(x,y),x)=0;
> eq2:=diff(f(x,y),y)=0;
> solve({eq1,eq2},{x,y});
```

Se han obtenido tres puntos críticos: $P_1(0, -1)$, $P_2\left(1, \frac{-3}{2}\right)$ y $P_3(2, -3)$.

Se calcula el valor del determinante de la matriz hessiana en cada uno de los puntos críticos:

```
> with(VectorCalculus): with(LinearAlgebra):
> Hessian(f(x,y), [x,y]);
> HES:=unapply(%, [x,y]);
> Determinant(HES(0,-1));
> Determinant(HES(1,-3/2));
> Determinant(HES(2,-3));
```

De lo anterior, se deduce que en $P_1(0, -1)$ y en $P_3(2, -3)$ la función tiene puntos de silla con valor:

```
> f(0,-1); f(2,-3);
```

Para decidir lo que ocurre en el otro punto se halla la derivada parcial segunda respecto de x :

```
> diff(f(x,y), x,x);
> DXX:=unapply(%, [x,y]);
> DXX(1,-3/2);
```

Por tanto, la función alcanza un mínimo relativo en el punto $P_2\left(1, \frac{-3}{2}\right)$ con valor:

```
> f(1,-3/2);
```

Para visualizar los resultados obtenidos, se puede pintar la función con sus puntos críticos:

```
> with(plots):
> P1:=pointplot3d([0,-1,-1], axes=framed, color=red, symbol=diamond):
> P2:=pointplot3d([1,-3/2,-1.25], axes=framed, color=black, symbol=diamond):
> P3:=pointplot3d([2,-3,-1], axes=framed, color=red, symbol=diamond):
> Gf:=plot3d(f(x,y), x=-1..3, y=0..-4, axes=framed, color=yellow):
> display(P1,P2,P3,Gf);
```

Ejemplo 6. Extremos condicionados (método de los multiplicadores de Lagrange)

El método de los multiplicadores de Lagrange se utiliza para hallar extremos absolutos de una función cuando las variables están sometidas a cierta condición o ligadura.

Por ejemplo, para hallar los extremos absolutos de la función $f(x,y) = x^3 + y^3$ sobre la circunferencia $x^2 + y^2 = 1$.

Se comienza visualizando la situación:

```
> restart;
> f:=(x,y)->x^3+y^3;
> g:=(x,y)->x^2+y^2-1;
> with(plots):
> func:=plot3d(f(x,y), x=-2..2, y=-2..2, color=blue):
> lig:=implicitplot3d(g(x,y)=0, x=-2..2, y=-2..2, z=-10..10, color=red):
> display({func,lig});
```

Se construye la función asociada a los multiplicadores de Lagrange:

```
> F:=(x,y,lambda)->f(x,y)+lambda*g(x,y);
```

y se obtienen sus puntos críticos:

```
> eq1:=diff(F(x,y,lambda),x)=0;
```

```
> eq2:=diff(F(x,y,lambda),y)=0;
```

```
> eq3:=diff(F(x,y,lambda),lambda)=0;
```

```
> raices:=solve({eq1,eq2,eq3},[x,y,lambda]);
```

```
> puntoscriticos:=allvalues(%);
```

Se halla el valor de la función en cada uno de los puntos asociados (sólo hay 6 diferentes) a los puntos críticos:

```
> f(0,1); f(0,-1); f(1,0); f(-1,0);
```

```
> evalf(f(sqrt(2)/2,sqrt(2)/2));
```

```
> evalf(f(-sqrt(2)/2,-sqrt(2)/2));
```

Por tanto, el máximo absoluto de la función es 1, que lo alcanza en los puntos (0,1) y (1,0), y el mínimo absoluto es -1, que lo alcanza en los puntos (0,-1) y (-1,0).

Ejercicio 1

Estudia la existencia de los siguientes límites y, en caso afirmativa, halla su valor:

a) $\lim_{(x,y) \rightarrow (0,0)} \frac{x+y}{x^2+y}$

b) $\lim_{(x,y) \rightarrow (0,0)} \frac{x}{x^2-y^2}$

c) $\lim_{(x,y) \rightarrow (0,0)} \frac{x+y}{x+y^3}$

Ejercicio 2

Halla las derivadas direccionales de las siguientes funciones en los puntos y direcciones que se indican:

a) $f(x,y) = 5 + x^2 - 3 \cdot y^2$, en el punto (1,2) y en la dirección $\theta = \frac{\pi}{6}$.

b) $g(x,y) = y^2 \cdot \sin(3xy)$, en el punto $(\pi,1)$ y en la dirección (3,-4).

Ejercicio 3

Halla el vector gradiente de la función $f(x,y,z) = x^2 + y^2 - 4z$, así como las direcciones de máximo y mínimo incremento de la función en el punto (2,-1,1). ¿Hay alguna dirección en que la derivada direccional sea nula?

Ejercicio 4

Halla la ecuación del plano tangente a:

a) El hiperboloide $z^2 - 2x^2 - 2y^2 = 12$ en el punto (1,-1,4).

b) La superficie $f(x,y) = (1 - x^2 - y^2)e^{2-x^2-y^2}$ en el punto (1,1,-1).

Ejercicio 5

Halla el polinomio de McLaurin de orden 2 de la función $f(x,y) = x \cos(y) + y \sin(x)$

Ejercicio 6

Halla los extremos relativos de las funciones: a) $f(x,y) = -x^3 + 4xy - 2y^2 + 1$ b) $g(x,y) = 1 - \sqrt[3]{x^2 + y^2}$

Ejercicio 7

Halla los extremos absolutos de la función $f(x,y) = \sin(xy)$ en $[0, \pi] \times [0, 1]$.

Ejercicio 8

Halla los extremos absolutos de la función $f(x,y) = xy$ sobre la circunferencia $x^2 + y^2 = 1$.

Ejercicio 9

Halla los extremos absolutos de la función $f(x,y,z) = x + y + z$ sobre el elipsoide $x^2 + 2y^2 + 4z^2 = 1$.